Journal of Avian Biology

JAV-00623
Low, M., Arlt, D., Pärt, T. and Öberg, M. 2015. Delayed timing of breeding as a cost of reproduction. – J. Avian Biol. doi: 10.1111/jav.00623

Supplementary material

Appendix 1

```
## JAGS (BUGS) code for Bayesian models in "Delayed Timing of Breeding as a Cost of Reproduction" ##
# Model code written by ML – September 2013

###############
### ARRIVAL ###
###############

#data
	#arr1 = arrival date in year 1
	#arr2 = arrival date in year 2
	#eggdat = standardized egg-laying date in year 1
	#flh = field layer height (short versus tall)
	#age = female age (young versus older)
	#yr = year
	#fail = birds who failed and didn't renest (1/0)
	#renest = birds who failed and renested (1/0)

model{

 #priors
	mu.alpha ~ dnorm(0,.0001)
	b.arr1 ~ dnorm(0,.0001)
	b.eggdat ~ dnorm(0,.0001)
	b.renest ~ dnorm(0,.0001)
	b.fail ~ dnorm(0,.0001)
	b.flh ~ dnorm(0,.0001)
	b.age ~ dnorm(0,.0001)
	tau ~ dgamma(.0001,.0001)
	tau.alpha ~ dgamma(.0001,.0001)

 #model
	for(i in 1:length(arr2)){

	arrival[i]<-alpha[yr[i]] +b.arr1*arr1[i] +b.eggdat*eggdat[i] +b.renest*renest[i] +b.fail*fail[i] +b.flh*flh[i]
	+b.age*age[i]

	arr2[i]~dnorm(arrival[i],tau) }

 #control for year effects
	for(k in 1:max(yr)){
	alpha[k] ~ dnorm(mu.alpha, tau.alpha) }

 #derive predictions
	pred.c<-mu.alpha +b.arr1*mean(arr1) +b.eggdat*mean(eggdat) +b.flh*mean(flh) +b.age*mean(age)

	pred.r<-mu.alpha +b.arr1*mean(arr1) +b.eggdat*mean(eggdat) +b.flh*mean(flh) +b.age*mean(age) +b.renest

	pred.f<-mu.alpha +b.arr1*mean(arr1) +b.eggdat*mean(eggdat) +b.flh*mean(flh) +b.age*mean(age) +b.fail

	renest.control.diff<-pred.r-pred.c
	fail.control.diff<-pred.f-pred.c
	renest.fail.diff<-pred.r-pred.f   } #close model
```

```
##############################
### Arrival-Breeding INTERVAL ###
##############################

#data
	#int.y = number days between arrival and egg laying
	#arr = arrival date in year 2
	#eggdat = standardized egg-laying date in year 1
	#flh = field layer height (short versus tall)
	#age = female age (young versus older)
	#yr = year
	#fail = birds who failed and didn't renest (1/0)
	#renest = birds who failed and renested (1/0)


model{
 #priors
	mu.alpha ~ dnorm(0,.0001)
	b.arr ~ dnorm(0,.0001)
	b.eggdat ~ dnorm(0,.0001)
	b.renest ~ dnorm(0,.0001)
	b.fail ~ dnorm(0,.0001)
	b.flh ~ dnorm(0,.0001)
	b.age ~ dnorm(0,.0001)
	tau ~ dgamma(.0001,.0001)
	tau.alpha ~ dgamma(.0001,.0001)

 #model
	for(i in 1:length(int.y)){

 interval[i]<-alpha[yr[i]] +b.arr*arr[i] +b.eggdat*eggdat[i] +b.renest*renest[i] +b.fail*fail[i] +b.flh*flh[i]
 +b.age*age[i]

 int.y[i]~dnorm(interval[i],tau) }

 #control for year effects
	for(k in 1:max(yr)){
	alpha[k] ~ dnorm(mu.alpha, tau.alpha) }

 #derive predictions
	pred.c<-mu.alpha +b.arr*mean(arr) +b.eggdat*mean(eggdat) +b.flh*mean(flh) +b.age*mean(age)

	pred.r<-mu.alpha +b.arr*mean(arr) +b.eggdat*mean(eggdat) +b.flh*mean(flh) +b.age*mean(age) +b.renest

	pred.f<-mu.alpha +b.arr*mean(arr) +b.eggdat*mean(eggdat) +b.flh*mean(flh) +b.age*mean(age) +b.fail

	renest.control.diff<-pred.r-pred.c
	fail.control.diff<-pred.f-pred.c
	renest.fail.diff<-pred.r-pred.f

 } #close model
```

```
############################
    ### EGG LAYING DATE ###
############################

#data
          #eld.y = egg laying date date in year 2
          #eggdat = standardized egg-laying date in year 1
          #flh = field layer height (short versus tall)
          #age = female age (young versus older)
          #yr = year
          #fail = birds who failed and didn't renest (1/0)
          #renest = birds who failed and renested (1/0)

model{
 #priors
          mu.alpha ~ dnorm(0,.0001)
          b.eggdat ~ dnorm(0,.0001)
          b.renest ~ dnorm(0,.0001)
          b.fail ~ dnorm(0,.0001)
          b.flh ~ dnorm(0,.0001)
          b.age ~ dnorm(0,.0001)
          tau ~ dgamma(.0001,.0001)
          tau.alpha ~ dgamma(.0001,.0001)

 #model
          for(i in 1:length(eld.y)){

          eld[i]<-alpha[yr[i]]+b.eggdat*eggdat[i]+b.renest*renest[i]+b.fail*fail[i]+b.flh*flh[i]+b.age*age[i]

          eld.y[i]~dnorm(eld[i],tau)        }

 #control for year effects
          for(k in 1:max(yr)){
          alpha[k] ~ dnorm(mu.alpha, tau.alpha) }

 #derive predictions
          pred.c<-mu.alpha+b.eggdat*mean(eggdat)+b.flh*mean(flh)+b.age*mean(age)
          pred.r<-mu.alpha+b.eggdat*mean(eggdat)+b.flh*mean(flh)+b.age*mean(age)+b.renest
          pred.f<-mu.alpha+b.eggdat*mean(eggdat)+b.flh*mean(flh)+b.age*mean(age)+b.fail

          renest.control.diff<-pred.r-pred.c
          fail.control.diff<-pred.f-pred.c
          renest.fail.diff<-pred.r-pred.f

 } #close model


############################
### FLEDGLINGS ZIP MODEL ###
############################

#data
          #fled.y = number of chicks fledged in year 2
          #eggdat = standardized egg-laying date in year 1
          #flh = field layer height (short versus tall)
          #age = female age (young versus older)
          #yr = year
          #fail = birds who failed and didn't renest (1/0)
          #renest = birds who failed and renested (1/0)

model{
 #priors
```

```
            mu.alpha ~ dnorm(0,.0001)
            b.eggdat ~ dnorm(0,.0001)
            b.renest ~ dnorm(0,.0001)
            b.fail ~ dnorm(0,.0001)
            b.flh ~ dnorm(0,.0001)
            b.age ~ dnorm(0,.0001)
            tau.alpha ~ dgamma(.0001,.0001)
            tau.psi ~ dgamma(.0001,.0001)
            mu.psi ~ dnorm(0,.0001)
            bpsi.eggdat ~ dnorm(0,.0001)
            bpsi.flh ~ dnorm(0,.0001)
            bpsi.nest ~ dnorm(0,.0001)
            bpsi.renest ~ dnorm(0,.0001)
            bpsi.fail ~ dnorm(0,.0001)

  #model
 #binomial part of ZIP controlling for true absences

for(i in 1:length(fled.y)){

  psi[i]<-ilogit(alpha.psi[yr[i]] +bpsi.eggdat*eggdat[i] +bpsi.flh*flh[i] +bpsi.renest*renest[i] +bpsi.fail*fail[i])

 w[i]~dbern(psi[i])

#count part of ZIP looking at true Poisson distribution

  fledged[i]<-exp(alpha[yr[i]] +b.eggdat*eggdat[i] +b.renest*renest[i] +b.fail*fail[i] +b.flh*flh[i] +b.age*age[i])

#combining Binomial (1/0) and Poisson (count)

 eff.lambda[i]<-fledged[i]*w[i]
  fled.y[i]~dpois(eff.lambda[i])
  }

  #control for year effects
        for(k in 1:max(yr)){
                alpha[k] ~ dnorm(mu.alpha, tau.alpha)
                alpha.psi[k] ~ dnorm(mu.psi, tau.psi) }

  #derive predictions for groups if egg laying date in year 2 is the same
   log(count.c)<-mu.alpha+b.eggdat*mean.egg+b.flh*mean.flh+b.age*mean.age
   log(count.r)<-mu.alpha+b.eggdat*mean.egg+b.flh*mean.flh+b.age*mean.age+b.renest
   log(count.f)<-mu.alpha+b.eggdat*mean.egg+b.flh*mean.flh+b.age*mean.age+b.fail

   zip.c<-ilogit(mu.psi+bpsi.eggdat*mean.egg+bpsi.flh*mean.flh)
   zip.r<-ilogit(mu.psi+bpsi.eggdat*mean.egg+bpsi.flh*mean.flh+bpsi.renest)
   zip.f<-ilogit(mu.psi+bpsi.eggdat*mean.egg+bpsi.flh*mean.flh+bpsi.fail)

  pred.c<-count.c*zip.c
  pred.r<-count.r*zip.r
  pred.f<-count.f*zip.f

  renest.control.diff<-pred.r-pred.c
  fail.control.diff<-pred.f-pred.c
  renest.fail.diff<-pred.r-pred.f

# derive predictions for groups if egg laying date in year 2 differ according to group-level differences in egg laying date
  log(count.c)<-mu.alpha+b.eggdat*mean.egg+b.flh*mean.flh+b.age*mean.age
  log(count.r)<-mu.alpha+b.eggdat*(mean.egg+1.05)+b.flh*mean.flh+b.age*mean.age+b.renest
  log(count.f)<-mu.alpha+b.eggdat*(mean.egg-3.32)+b.flh*mean.flh+b.age*mean.age+b.fail

  zip.c<-ilogit(mu.psi+bpsi.eggdat*mean.egg+bpsi.flh*mean.flh)
```

```
 zip.r<-ilogit(mu.psi+bpsi.eggdat*(mean.egg+1.05)+bpsi.flh*mean.flh+bpsi.renest)
 zip.f<-ilogit(mu.psi+bpsi.eggdat*(mean.egg-3.32)+bpsi.flh*mean.flh+bpsi.fail)

 pred.c<-count.c*zip.c
 pred.r<-count.r*zip.r
 pred.f<-count.f*zip.f

 renest.control.diff<-pred.r-pred.c
 fail.control.diff<-pred.f-pred.c
 renest.fail.diff<-pred.r-pred.f

 } #close model


#####################################
### ADULT RETURN RATES (SURVIVAL) ###
#####################################

#data
	#return.y = was adult seen in following year (1/0)
	#flh = field layer height (short versus tall)
	#age = female age (young versus older)
	#yr = year
	#fail = birds who failed and didn't renest (1/0)
	#renest = birds who failed and renested (1/0)

model{

#priors
	mu.alpha ~ dnorm(0,.0001)
	b.renest ~ dnorm(0,.0001)
	b.age ~ dnorm(0,.0001)
	b.flh ~ dnorm(0,.0001)
	tau.alpha ~ dgamma(.0001,.0001)

#model

for(i in 1:length(return.y)){
	survival[i]<-ilogit(alpha[yr[i]]+b.renest*renest[i]+b.age*age[i]+b.flh*flh[i])
	return.y[i]~dbern(survival[i])
}

#random effects of yr
 for(k in 1:max(yr)){
	alpha[k]~dnorm(mu.alpha, tau.alpha)
}

#predictions

pred.c<-ilogit(mu.alpha+b.age*mean(age)+b.flh*mean(flh))
pred.r<-ilogit(mu.alpha+b.renest+b.age*mean(age)+b.flh*mean(flh))
diff<-pred.c-pred.r

} # close model
```